
1. Введение

Сортировку  следует  понимать  как  процесс  перегруппировки  заданного  множества  объектов  в  некотором  определенном  порядке. Цель  сортировки – облегчить последующий  поиск  элементов в  таком  отсортированном  множестве. Мы  встречаемся  с  отсортированными  объектами  в  телефонных  книгах, в  списках  подоходных  налогов, в  оглавлениях  книг, в  библиотеках, в словарях, на  складах – почти  везде, где  нужно  искать  хранимые  объекты. Сортировка применяется во всех без исключения областях программирования, будь то базы данных или математические программы.

Практически каждый алгоритм сортировки можно разбить на три части:

- сравнение, определяющее упорядоченность пары элементов;

- перестановку, меняющую местами пару элементов;

- собственно сортирующий алгоритм, который осуществляет сравнение и перестановку элементов до тех пор, сока все элементы множества не будут упорядочены.

Выбор  алгоритма  зависит  от  структуры  обрабатываемых  данных. В  случае  сортировки  соответствующие  методы  разбиты  на  два  класса – сортировку  массивов  и  сортировку  файлов. Иногда  их  называют  внутренней  и  внешней  сортировкой, поскольку  массивы  хранятся  в  оперативной, внутренней  памяти  машины  со  случайным  доступом, а  файлы  обычно  размещаются  в  более  медленной, но  и  более  емкой  внешней  памяти.
В своей работе я рассмотрю алгоритмы сортировки массивов. Массив - это проиндексированная, упорядоченная последовательность однотипных элементов. В данном случае под элементами подразумеваются числа. Сортировка массива - это процесс, направленный на упорядочение массива. Полная сортировка выстраивает элементы массива по возрастанию или по убыванию. То есть каждый последующий элемент больше либо равен предыдущему (X1>X2>X3>…>XN) или наоборот(X1<X2<X3<…<XN). 

Каждый алгоритм сортировки имеет  и  свои  достоинства,  и  свои  недостатки,  и  выбирать  алгоритм  нужно,  исходя из  конкретной  постановки  задачи.
2.Основная часть

2.1.Сортировка пузырьком

Сортировка пузырьком (BubbleSort) - один из простейших методов сортировки. Название метода отражает его сущность: на каждом шаге самый «легкий» элемент поднимается до своего места («всплывает»). Для   этого   мы   просматриваем   все элементы снизу вверх, берем пару соседних элементов и, в случае, если они стоят неправильно, меняем их местами.

Вместо поднятия самого «легкого» элемента можно «топить» самый «тяжелый».

Т.к. за каждый шаг на свое место встает ровно  1 элемент (самый  «легкий»  из оставшихся), то нам потребуется выполнить N шагов.

Текст процедуры сортировки можно записать так:

procedure bubble_sort(var a:list; n : integer);

var

    i, j, k : integer;

begin

   for i := 0 to n-1 do

       for j := n-1 downto i do

       if (a[j-1] > a[j]) then begin

           k := a[j-1];

           a[j-1] := a[j];

           a[j] := k;

        end;

end;

Для передачи массивов в процедуры и функции, необходимо определить новый тип:

type

   list = array[0..10000] of integer;

который избавляет от необходимости писать длинные команды и уменьшает количество возможных опечаток и ошибок.

Алгоритм не использует дополнительной памяти, т.е. все действия осуществляются на одном и том же массиве.

Сложность алгоритма сортировки пузырьком составляет O(N2), количество операций сравнения: 
N ( (N-1)/2. Это очень плохая сложность, по алгоритм имеет два плюса.

Во-первых, он легко реализуется, а значит, может и должен применяться в тех случаях, когда требуется однократная сортировка массива. При этом размер массива не должен быть больше 10000, т.к. иначе алгоритм сортировки пузырьком не будет укладываться в отведенное время.

Во-вторых, сортировка пузырьком использует только сравнения и перестановки соседних элементов, а значит, может использоваться в тех задачах, где явно разрешен только такой обмен и для сортировки, например, списков.

Существуют разнообразные «оптимизации» сортировки пузырьком, которые усложняют (а нередко и увеличивают время работы алгоритма), но не приносят выгоды ни в плане сложности, ни в плане быстродействия. На этом плюсы сортировки пузырьком заканчиваются.
2.2.Сортировка прямым выбором

Рассмотрим еще один квадратичный алгоритм, который, однако, является оптимальным по количеству присваиваний и может быть использован, когда по условию задачи необходимо явно минимизировать количество присваиваний.

Суть метода заключается в следующем: мы будем выбирать минимальный элемент в оставшейся части массива и приписывать его к уже отсортированной части. Повторив эти действия N раз, мы получим отсортированный массив.

procedure select_sort(var a : list; n : integer);

var

   i, j, k : integer;

begin

   for i := 0 to n-1 do begin

       k := i;

       for j := i+1 to n-1 do

            if (a[j]<a[k]) then k := j;

       j := a[k];

       a[k] := a[i];

       a[i] := j;

    end;

end;

Количество сравнений составляет O(N2), а количество присваиваний всего O(N). В целом это плохой метод, и он должен быть использован только в случаях, когда явно необходимо минимизировать количество присваиваний.

2.3. Пирамидальная сортировка

Одним из эффективных алгоритмов сортировки является пирамидальная сортировка (работает за O(N logN), в которой используются понятие кучи.

Куча – это структура данных, которая может выдавать минимальный (или максимальный) элемент за О(1), добавление нового элемента или добавление минимального элемента происходит О(logN) (где N – количество элементов в куче). Другие операции над кучей не определены (хотя при необходимости могут быть введены, однако эффективность их будет невысокой, и они обязаны поддерживать свойства кучи). 

Мы будем выбирать из кучи самый большой элемент, и записывать его в начало уже отсортированной части массива (сортировка выбором в обратном порядке). Т.е. отсортированный массив будет строиться от конца к началу. Такие ухищрения необходимы, чтобы не было необходимости в дополнительной памяти и для ускорения работы алгоритма — куча будет располагаться в начале массива, а отсортированная часть будет находиться после кучи.

Напомним свойство кучи максимумов: элементы с индексами i*2 + 1 и i*2 + 2 не больше, чем элемент с индексом i (естественно, если i*2 + 1 и i*2 + 2 лежат в пределах кучи). Пусть n — размер кучи, тогда вторая половина массива (элементы от n/2 + 1 до n) удовлетворяют свойству кучи. Для остальных элементов вызовем процедуру «проталкивания» по куче, начиная с n/2 до 0.

procedure down_heap(var a : list; k : integer; n : integer);

var

    y, temp : integer;

begin

     temp := a[k];

     while (k*2+1 < n) do begin

         y := k*2+1;

         if (y < n-1) and (a[y] < a[y+1]) then inc(y);

         if (temp >= a[y]) then break;

         a[k] := a[y];

         k := y;

    end;

    a[k] := temp;

end;

Эта процедура получает указатель на массив, номер элемента, который необходимо протолкнуть и размер кучи. У нее есть небольшие отличия от обычных процедур работы с кучей. Номер минимального предка хранится в переменной у, если необходимость в обменах закончена, то мы выходим из цикла и записываем просеянную переменную на предназначенное ей место.

Сама сортировка будет состоять из создания кучи из массива и N переносов элементов с вершины кучи с последующим восстановлением свойства кучи:

procedure heap_sort(var a : list; n : integer);

var

i, temp : integer;

begin

for i := n div 2 - 1 downto 0 do down_heap(a, i, n);

for i := n-1 downto 0 do begin

temp := a[i];

a[i] := a[0];

a[0] := temp;

down_heap(a, 0, i);

end;

end;

Всего в процессе работы алгоритма будет выполнено 3×N/2-2 вызова процедуры down_heap, каждый из которых занимает O(logN). Таким образом, мы и получаем искомую сложность в O(N logN), не используя при этом дополнительной памяти. Количество присваиваний также составляет O(N logN).
Пирамидальную сортировку следует осуществлять, если из условия задачи понятно, что единственной разрешенной операцией является «проталкивание» элемента по куче, либо в случае отсутствия дополнительной памяти.

2.4. Быстрая сортировка.

Этот  улучшенный  метод  сортировки  основан  на  обмене. Это  самый  лучший  из  всех  известных  на  данный  момент  методов сортировки  массивов. Был разработан в 1962 г. Его  производительность  столь  впечатляюща, что  изобретатель  Ч. Хоар  назвал  этот  метод  быстрой   сортировкой  (Quicksort).
Выбирается некий опорный элемент и все числа, меньшие его перемещаются в левую часть массива, а все числа большие его — в правую часть. Затем вызываем процедуру сортировки для каждой из этих частей.

Таким образом, процедура сортировки должна принимать указатель на массив и две переменные, обозначающие левую и правую границу сортируемой области.

Остановимся более подробно на выборе опорного элемента. В некоторых книгах рекомендуется выбирать случайный элемент между левой и правой границей. Хотя теоретически это красиво и правильно, но на практике следует учитывать, что функция генерации случайного числа достаточно медленная и такой метод заметно ухудшает производительность алгоритма в среднем.

Наиболее часто используется середина области, т.е. элемент с индексом (l + r)/2. При таком подходе используются быстрые операции сложения и деления на два, и в целом он работает достаточно неплохо. Однако в некоторых задачах, где сутью является исключительно сортировка, хитрое жюри специально подбирает тесты так, чтобы «завалить» стандартную быструю сортировку с выбором опорного элемента из середины. Стоит заметить, что это очень редкая ситуация, но все же стоит знать, что можно выбирать произвольный элемент с индексом т так, чтобы выполнялось неравенство l ≤ т ≤ r. Чтобы это условие выполнялось, достаточно выбрать произвольные два числа х и у и выбирать т исходя из следующего соотношения: 

т = (х (  l + у ( r)/(х + у).

В целом такой метод будет незначительно проигрывать выбору среднего элемента, т.к. требует двух дополнительных умножений.

Приведем текст процедуры быстрой сортировки с выбором среднего элемента в качестве опорного:

procedure quick_sort(var a : list; left, right : integer);

var

   i, j, temp, p : integer;

begin

    i := left;

    j := right;

    p := a[(left+right) div 2];

    repeat

       while (a[i] < p) do inc(i);

       while (a[j] > p) do dec(j);

       if (i <= j) then begin

          temp := a[i];

          a[i] := a[j];

          a[j] := temp;

          inc(i);

          dec(j);

       end;

     until (i > j);

     if (j > left) then quick_sort(a, left, j);

     if (i < right) then quick_sort(a, i, right);

end;

Чтобы воспользоваться быстрой сортировкой, необходимо передать в процедуру левую и правую границы сортируемого массива, (т.е., например, вызов для массива а будет выглядеть как quick_sort(a, 0, n-1).

Алгоритм быстрой сортировки в среднем использует O(N logN) сравнений и O(N logN) присваиваний (на практике даже меньше) и использует O(logN) дополнительной памяти (стек для вызова рекурсивных процедур). В худшем случае алгоритм имеет сложность O(N2) и использует О(N) дополнительной памяти, однако вероятность возникновения худшего случая крайне мала: на каждом шаге вероятность худшего случая равна 2/N, где N — текущее количество элементов.

Рассмотрим возможные оптимизации метода быстрой сортировки.

Во-первых, при вызове рекурсивной процедуры возникают накладные расходы на хранение локальных переменных (которые нам не особо нужны при рекурсивных вызовах) и другой служебной информацией. Таким образом, при замене рекурсии стеком мы получим небольшой прирост производительности и небольшое снижение требуемого объема дополнительной памяти.

Во-вторых, как мы знаем, вызов процедуры — достаточно накладная операция, а для небольших массивов быстрая сортировка работает не очень хорошо. Поэтому, если при вызове процедуры сортировки в массиве находится меньше, чем К элементов, разумно использовать какой-либо нерекурсивный метод, например, сортировку вставками или выбором. Число К при этом выбирается в районе 20, конкретные значения подбираются опытным путем. Такая модификация может дать до 15% прироста производительности.

Быструю сортировку можно использовать и для двусвязных списков (т.к. в ней осуществляется только последовательный доступ с начала и с конца), но в этом случае возникают проблемы с выбором опорного элемента — его приходится брать первым или последним в сортируемой области. Эту проблема можно решить неким набором псевдослучайных перестановок элементов списка, тогда даже если данные были подобраны специально, эффект нейтрализуется.

2.5.Сортировка слиянием

 Сортировка слияниями также основывается на идее, которая уже была нами затронута при рассмотрении алгоритма поиска двух максимальных элементов. В этом алгоритме мы сначала разобьем элементы на пары и упорядочим их внутри пары. Затем из двух пар создадим упорядоченные четверки и т.д.

	3
	7
	8
	2
	1
	6
	1
	5
	Последовательности длины 1

	37
	28
	46
	1 5
	Слияние до упорядоченных пар

	2378
	1456
	Слияние пар в упорядоченные четверки

	12345678
	Слияние пар в упорядоченные четверки


Интерес представляет сам процесс слияния: для каждой из половинок мы устанавливаем указатели на начало, смотрим, в какой из частей элемент по указателю меньше, записываем этот элемент в новый массив и перемещаем соответствующий указатель.

Опишем процедуру слияния следующим образом:

procedure merge(var a, b : list; c, d, e : integer);

var

   p1, p2, pres : integer;

begin

  p1 := c;

  p2 := d;

  pres := c;

  while (p1 < d) and (p2 < e) do

    if (a[p1] < a[p2]) then begin

    b[pres] := a[p1];

    inc(pres);

    inc(p1);

  end else begin

    b[pres] := a[p2];

    inc(pres);

    inc(p2);

  end;

  while (p1 < d) do begin

    b[pres] := a[p1];

    inc(pres);

    inc(p1);

  end;

   while (p2 < e) do begin

     b[pres] := a[p2];

     inc(pres);

     inc(p2);

   end;

end;

Здесь а – исходный массив, b – массив результата, c и d – указатели на начало первой и второй части соответственно, е – указатель на конец второй части.

Далее опишем нерекурсивную процедуру сортировки слиянием.

procedure merge_sort(var a : list; n : integer);

var

  c, k, d, e : integer;

  temp, a2, b : ^list;

  temp_list : list;

begin

   b := @temp_list;

   a2 := @a;

   k := 1;

   while (k <= 2*n) do begin

   c := 0;

   while (c < n) do begin

      if (c + k < n) then

     d := c + k

     else

       d := n;

    if (c + 2*k < n) then

       e := c + 2*k

    else

      e := n;

      merge(a2^, b^, c, d, e);

      c := c + k*2;

   end;

   temp := a2;

   a2 := b;

   b := temp;

   k := k*2;

 end;

 for c := 0 to n-1 do

    a[c] := a2[c];

end;

Рекурсивная реализация сортировки слияниями несколько проще, но обладает меньшей эффективностью и требует O(logN)  дополнительной памяти.

Алгоритм имеет сложность O(N logN) и требует O(N)  дополнительной памяти.

В оригинале этот алгоритм был придуман для сортировки данных во внешней памяти (данные были расположены в файлах) и требует только последовательного доступа. Этот алгоритм применим для сортировки односвязных списков.
3. Сравнение производительности сортировок.

	Название
	Сравнений
	Присваиваний
	Память

	Пузырьком
	O(N2)
	O(N2)
	0

	Выбором
	O(N2)
	O(N)
	0

	Пирамидальная
	O(N logN)
	O(N logN)
	0

	Быстрая
	O(N logN)
	O(N logN)
	O( logN)

	Слиянием
	O(N logN)
	O(N logN)
	O(N)


4. Заключение
Анализируя сортировки массивов, я пришел к следующим выводам. Сортировка пузырьком является простой в реализации, но крайне медленной. Сортировка прямым выбором сравнима по скорости с пузырьковой, но выигрывает по количеству присваиваний. Из представленных сортировок, на практике быстрее всех работает быстрая сортировка. Она является оптимальной по простоте написания и надежной в работе. Пирамидальная сортировка хороша тем, что не требует лишней памяти.

